/**		random.d - Random numbers generator

// Written in the D programming language 1.0

Authors:	Wei Li <oldrev@gmail.com> (in D)
			and below
License:	BSD/MIT
*/


// System.Random.cs
//
// Authors:
//   Bob Smith (bob@thestuff.net)
//   Ben Maurer (bmaurer@users.sourceforge.net)
//
// (C) 2001 Bob Smith.  http://www.thestuff.net
// (C) 2003 Ben Maurer
//

//
// Copyright (C) 2004 Novell, Inc (http://www.novell.com)
//
// Permission is hereby granted, free of charge, to any person obtaining
// a copy of this software and associated documentation files (the
// "Software"), to deal in the Software without restriction, including
// without limitation the rights to use, copy, modify, merge, publish,
// distribute, sublicense, and/or sell copies of the Software, and to
// permit persons to whom the Software is furnished to do so, subject to
// the following conditions:
// 
// The above copyright notice and this permission notice shall be
// included in all copies or substantial portions of the Software.
// 
// THE SOFTWARE IS PROVIDED "AS IS", WITHOUT WARRANTY OF ANY KIND,
// EXPRESS OR IMPLIED, INCLUDING BUT NOT LIMITED TO THE WARRANTIES OF
// MERCHANTABILITY, FITNESS FOR A PARTICULAR PURPOSE AND
// NONINFRINGEMENT. IN NO EVENT SHALL THE AUTHORS OR COPYRIGHT HOLDERS BE
// LIABLE FOR ANY CLAIM, DAMAGES OR OTHER LIABILITY, WHETHER IN AN ACTION
// OF CONTRACT, TORT OR OTHERWISE, ARISING FROM, OUT OF OR IN CONNECTION
// WITH THE SOFTWARE OR THE USE OR OTHER DEALINGS IN THE SOFTWARE.
//


module dotmars.base.random;

import dotmars.base.math;

version (Windows)
{
	extern(Windows) int QueryPerformanceCounter(ulong *count);
	extern(Windows) uint GetTickCount();
}

version (Posix)
{
	import dotmars.runtime.stdclib.time;
}



final public class Random
{
	const int MBIG = int.max;
	const int MSEED = 161803398;
	const int MZ = 0;

	int inext, inextp;
	int [56] SeedArray;

	public this ()
	{
		version(Win32)
			this(cast(int)(GetTickCount()));

		version(Posix)
			this(cast(int)(clock()));
	}

	public this (int Seed)
	{
		int ii;
		int mj, mk;

		// Numerical Recipes in C online @ http://www.library.cornell.edu/nr/bookcpdf/c7-1.pdf
		mj = MSEED - Math.abs (Seed);
		SeedArray [55] = mj;
		mk = 1;
		for (int i = 1; i < 55; i++) {  //  [1, 55] is special (Knuth)
			ii = (21 * i) % 55;
			SeedArray [ii] = mk;
			mk = mj - mk;
			if (mk < 0)
				mk += MBIG;
			mj = SeedArray [ii];
		}
		for (int k = 1; k < 5; k++) {
			for (int i = 1; i < 56; i++) {
				SeedArray [i] -= SeedArray [1 + (i + 30) % 55];
				if (SeedArray [i] < 0)
					SeedArray [i] += MBIG;
			}
		}
		inext = 0;
		inextp = 31;
	}

	protected double sample ()
	{
		int retVal;

		if (++inext  >= 56) inext  = 1;
		if (++inextp >= 56) inextp = 1;

		retVal = SeedArray [inext] - SeedArray [inextp];

		if (retVal < 0) retVal += MBIG;

		SeedArray [inext] = retVal;

		return retVal * (1.0 / MBIG);
	}

	public int next () {
		return cast(int)(sample * int.max);
	}

	public int next (int maxValue)
	in {
		assert(maxValue >= 0);
	}
	body {
		return cast(int)(sample () * maxValue);
	}

	public int next (int minValue, int maxValue)
	in {
		assert(minValue <= maxValue);
	}
	out(result) {
		assert(result >= minValue && result <= maxValue);
	}
	body {
		uint diff = cast(uint)(maxValue - minValue);
		if (diff == 0)
			return minValue;

		int result = cast(int)(sample  * diff + minValue);
		return ((result != maxValue) ? result : (result - 1));
	}

	public void nextBytes (ubyte [] buffer)
	in {
		assert(buffer !is null);
		assert(buffer.length > 0);
	}
	body {
		for (int i = 0; i < buffer.length; i++) {
			buffer [i] = cast(ubyte)(sample  * (ubyte.max + 1)); 
		}
	}

	public double nextDouble () {
		return sample;
	}
}
